Determination of Deposition Probability on Human Form

Eugene Yee
DRDC Suffield Research Centre
5 April 2017

Deposition Probability

Cloud transport and diffusion models have evolved substantially in recent years, making available such tools as the HPAC model which is well-suited for predicting the large-scale transport and diffusion of toxic agent material in the atmosphere. However, for small-scale problems such as the deposition onto the human form, it is appropriate to incorporate local transport effects attributable to droplet mass (or, size) as well as other processes such as aerodynamic drag. The information is embodied in the collection (or trapping) efficiency of the human form when placed in the path of a challenge liquid (or solid) aerosol stream, defined simply as the amount of particulates of a given size (diameter) actually impacting on any portion of the human body relative to the amount which would pass through its projected area if it were removed. The calculations required to determine the trapping efficiency of the human form for a challenge aerosol concentration are described in this note.

Digital Model of Human Form

To calculate the deposition probability, we need a realistic and accurate representation of the human form. To that end, the geometry of a computerized mannikin used as a representation of the actual human form corresponds to a realistic and accurate representation of a nude standing human male with arms spread (see figure below).

```
Import["C:\\tmp\\Model of Human Form.png"]
```

Figure 1: Digital representation of the human form.

This male figure has a height of 1.8 m, with a corresponding body surface area of 1.85 m^2. The geometry of the human form is extremely complex and, as a result, a high-fidelity digitization of the form is a challenging problem. The accuracy of the detailed modeling of the disturbed flow around the human form depends critically on the quality of this digitization (discrete representation). To this purpose, the surface of the modeled human form has been represented using 89,360 nodes (points) and a corresponding connectivity list. This list provides information on how these nodes are connected together to form 178,716 triangular elements (or, patches) that provide an accurate representation for the surface of the human body.

Computation of Shear Velocity on Surface of Human Form

The first step in the calculation of the deposition probability of the human form involves determination of the distribution of the shear (or friction) velocity u_* on the surface of the human body. Here, the shear velocity is defined as $u_* = \sqrt{\tau_w / \rho}$ where τ_w is the wall shear stress and ρ is the mass density of the fluid (air in our case). The calculation of the distribution of the shear velocity on the surface of the human form requires a knowledge of the fluid phase velocity around the human form. Indeed, a proper representation of the highly disturbed complex flow around the human form is crucial for the accurate prediction of the deposition rate of particulates on the surface of the human body.

To determine the distribution of the shear velocity on the surface of the human form, we use computational fluid dynamics to predict the highly disturbed flow around the human body. To this purpose, we use the Reynolds-averaged Navier-Stokes (RANS) equations, in conjunction with a conventional k-ϵ turbulence closure model for this purpose. Practically speaking, this strategy is computationally efficient and is to be preferred for the prediction of the flow around the human body. In RANS, the flow variables (such as velocity and pressure) are decomposed into a mean component and a fluctuating component. The mean velocity field and the pressure are obtained by solving the continuity and RANS equations, whereas the turbulence closure model is used to obtain the statistics of the fluctuating velocity which is not resolved explicitly in the computations. A standard wall function is use as the boundary condition on the human body surface for turbulent flows. Accordingly, the "law of the wall" for the turbulent streamwise mean velocity profile is imposed at the adjacent grid point from a location on the (solid) body surface. That is,

$$u^{+*} = n^{+*} \text{ for } n^{+*} < 11.225, \tag{1}$$

$$u^{+*} = \frac{1}{\kappa} \log\left(E\, n^{+*}\right) \text{ for } n^{+*} > 11.225, \tag{2}$$

where "+*" denotes a quantity expressed in wall units; u^{+*} and n^{+*} are the non-dimensional velocity and normal distance from the wall (body surface); and, κ = 0.42 (von Karman's constant) and E = 9.81. Note that the $u^{+*} = \overline{v_t}/u_*$ where $\overline{v_t}$ is the mean velocity parallel to the wall (body surface).

The figure below displays an example of the computation of the highly disturbed velocity field around and in the vicinity of the human form for the case of the body oriented at an angle of 0° with respect to the incident wind direction.

```
Import["C:\\tmp\\Flow Around Human Form.png"]
```



Figure 2: RANS methodology used in the prediction of the highly disturbed velocity field in and around the human form. The body is oriented at 0° with respect to the incoming (prevailing) wind direction (along the positive *x* axis in this example).

For visualization purposes, the mean flow fields and streamlines in two cross-sections through the human form are shown in the figure above: namely, in a vertical *x*-*z* plane bisecting the human form along a vertical plane through the body and in a horizontal *x*-*y* plane bisecting the human form at half-height. In particular, stream traces (top panels) and the velocity field components (bottom panels) along a vertical and a horizontal cross-section through the human form are exhibited in this figure. Superimposed in the various panels of this figure is the velocity magnitude, which has been false colored.

From Figure 2 above, it can be seen that complicated vortices and swirling flows are generated around and behind the human form. Of these, the streamline traces (top panels) clearly exhibit the complexity of the swirling flow and the three-dimensional vortical patterns that are generated behind the human body. There are stagnation points on the windward side of the body. The side views of the human form show the presence of two vertically stacked large counter-clockwise rotating eddies generated in the wake region of the human form. Indeed, there is a fairly large recirculation zone present in the rear of the body. This zone exhibits a strong reverse flow to the incoming (prevailing) flow. The presence of

these flow features implies that any aerosol droplets (particles) that are transported downstream in the immediate vicinity of the body can potentially be entrained into the recirculating flow in the body wake region and deposited subsequently in the rear region of the body.

Calculation of the Distribution of Deposition Velocity on the Human Form

Once the distribution of the shear velocity on the surface of the human form has been determined, this information can be used to calculate the distribution of the deposition velocity v_d on the human body surface. The non-dimensional deposition velocity for particles released with a uniform concentration C_0 near the surface is given by

$$v_d^+ = \frac{J}{C_0 u_*} , \tag{3}$$

where J is the particle mass flux to the wall per unit time and u_* is the flow shear velocity.

There are a number of semi-empirical model relationships for the non-dimensional deposition velocity v_d^+. A simple empirical equation for the non-dimensional deposition velocity suggested by Wood (1981) is given as

$$v_d^+ = 0.057\, Sc^{-2/3} + 4.5 \times 10^{-4}\, \tau^{+2} + u_t^+ , \tag{4}$$

where Sc = ν/D is the Schmidt number with D being the mass diffusivity of an aerosol particle of diameter D_p (cm) given as

$$D = \frac{k_B\, T C_c}{3\, \pi \eta D_p} , \tag{5}$$

where k_B is Boltzmann's constant (erg K^{-1}), T is the temperature (K), η is the dynamic viscosity of air (poise or g $cm^{-1}\, s^{-1}$) and C_c is the Cunningham slip correction factor. The Cunningham slip correction factor C_c is determined from the following equation:

$$C_c = 1 + 1.246\, \frac{\lambda}{r_p} + 0.42\, \frac{\lambda}{r_p} \exp\left(-0.87\, \frac{r_p}{\lambda}\right) , \tag{6}$$

where λ is the mean free path of air molecules (cm) and $r_p = D_p/2$ is the particle radius. Some relevant physical constants are: k_B = 1.38 × 10^{-16} erg K^{-1}; η = 1.83 × 10^{-4} poise; and λ = 0.653 × 10^{-5} cm at atmospheric pressure. In Eq. (4), τ^+ is the non-dimensional particle relaxation time defined as

$$\tau^+ = \frac{S\, D_p^2\, u_*^2}{18\, \nu^2}\, C_c , \tag{7}$$

where S ≡ ρ_p/ρ (ρ_p is the particle density (g cm^{-3}) and ρ is the fluid (air) density (g cm^{-3})) is the particle-to-fluid density ratio and $\nu \equiv \eta/\rho$ is the kinematic viscosity of the fluid ($cm^2\, s^{-1}$). For air, ν = 0.1461 $cm^2\, s^{-1}$. Finally, in Eq. (4), the last term accounts for the contribution to particle deposition velocity by gravitational sedimentation on horizontal surface, which is defined as

$$u_t^+ = \tau^+ g^+ , \tag{8}$$

where g^+ is expressed with the following form:

$$g^+ = \frac{\nu}{u_*^3}\, g . \tag{9}$$

The acceleration due to gravity has the value $g = 980.665$ cm s^{-2}.

Note that the model of Wood (1981) predicts rather large values for the normalized deposition velocity for large values of the non-dimensional particle relaxation time τ^+. To remove this unphysical aspect of the model, we place an upper bound on possible values for the normalized deposition velocity. To this purpose, we apply an upper bound of 0.14 on the normalized deposition velocity in agreement with available data sets on the deposition velocity for horizontal and vertical surfaces. However, given some preliminary experience with the model, this upper bound is perhaps not needed with respect to its application to the human form.

Reference: Wood, N. B. (1981). A simple method for the calculation of turbulent deposition to smooth and rough surfaces. *Journal of Aerosol Science*, 12, 275-290.

Calculation of the Deposition Probability

Once the deposition velocity has been determined, the probability of deposition p_d can be calculated. The non-dimensional deposition velocity $v_d^+ = v_d/u_*$ (v_d is the deposition velocity) and $p_d = v_d/U$ where U is the reference wind speed in the inlet plane (at body height). Consequently, the probability of deposition can be calculated from the non-dimensional deposition velocity as follows:

$$p_d = \frac{u_* v_d^+}{U} \,, \tag{10}$$

or, equivalently,

$$p_d = \frac{v_d}{U} \,. \tag{11}$$

Implementation of Model

In this section, we provide an implementation of the model described in the previous section.

- Define some physical constants required by the penetration model for material system

```
k_B = 1.38 × 10^-16 ;          (* Boltzmann's constant in erg K^-1 *)
T = 293.0 ;                    (* air temperature in K *)
λ = 0.653 × 10^-5 ;
(* mean free path of air molecules at atmospheric pressure in cm *)
η = 1.83 × 10^-4 ;             (* dynamic viscosity of air in poise (g cm^-1 s^-1) *)
ρ = 0.001225;                  (* density of air in g cm^-3 *)
ρ_P = 1.0 ;                    (* density of particles in g cm^-3 *)
ν = η / ρ;                     (* kinematic viscosity of air in cm^2 s^-1 *)
SRatio = ρ_P / ρ               (* particle-to-fluid density ratio *)
g = 980.665                    (* acceleration due to gravity in cm s^-2 *)
```

```
816.327
```

```
980.665
```

- Function to compute the Cunningham slip correction factor

```
(* Cunningham slip correction *)
(* Dp is the particle diameter in cm *)
Cc[Dp_] := 1.0 + 1.246 × 2 λ/Dp + 0.42 × 2 λ/Dp Exp[-0.87 Dp/(2 λ)]
```

- Function to compute the diffusion coefficient *D* of the particle

```
(* Diffusion coefficient D in cm^2 s^-1 *)
(* Dp is the particle diameter in cm *)

DiffCoef[Dp_] := (k_B T Cc[Dp]) / (3.0 Pi η Dp)
```

- Function to compute Schmidt number Sc

```
(* Schmidt number Sc *)
(* ν is the kinematic visocity of air in cm^2 s^-1 *)
(* Dp is the particle diameter in cm *)
(* DiffCoef is diffusion coefficient or diffusivity of the particle in cm^2 s^-1 *)

Sc[Dp_] := ν / DiffCoef[Dp]
```

- Function to compute the non-dimensionalized particle relaxation time τ^+

```
(* Non-dimensionalized particle relaxation time τ^+ (tauPlus) *)
(* Dp is the particle diameter in cm *)
(* ustar is the shear or friction velocity
 of surface where deposition is to occur in cm s^-1 *)

tauPlus[Dp_, ustar_] := ((SRatio Dp^2 ustar^2) / (18.0 ν^2)) Cc[Dp]
```

- Function to compute contribution to the deposition velocity due to gravitational sedimentation on horizontal surface u_t

```
(* Contribution arising from gravitational sedimentation on horizontal surface *)
(* Dp is the particle diameter in cm *)
(* ustar is the shear or friction velocity
 of surface where deposition is to occur in cm s^-1 *)

ut[Dp_, ustar_] := tauPlus[Dp, ustar] ν g / (ustar^3)
```

- Function to compute the non-dimensionalized deposition velocity $v_d^+ \equiv v_d/u_*$ for vertical surface with gravity in the flow direction (one needs to add the term u_t for horizontal surface with gravity perpendicular to the flow direction)

```
(* Non-dimensionalized deposition velocity vdPlus *)
(* Dp is the particle diameter in cm *)
(* ustar is the shear or friction velocity
 of surface where deposition is to occur in cm s^-1 *)

vdPlus[Dp_, ustar_] := Min[0.14, 0.057 Sc[Dp]^(-2/3) + 4.5 × 10^(-4) tauPlus[Dp, ustar]^2 ]
```

Example Application of the Model for Deposition Velocity

As an example of the use of the model, consider case of fully developed flow over a vertical surface where the shear velocity is u_* = 34.2 cm s^{-1}. Consider the particle size (diameter) range from 0.01 × 10^{-4} cm to 50.0 × 10^{-4} cm. First, let us calculate the non-dimensional particle relaxation time τ^+ for this range of particle sizes.

```
(* Set the friction velocity in cm s^-1 *)
ustar = 34.2;

(* Calculate particle diameters, logarithmically space from 0.01 to 50 μm *)

log10D = Range[-2, 2, 0.05];
partDiam = 10^log10D × 10^-4 ;  (* particle diameters in cm *)

(* For the range of particle diameters given,
compute the non-dimensional particle relaxation times τ^+ *)

tauPlusV = Table[tauPlus[partDiam[[i]], ustar], {i, 1, Length[partDiam]}];

(* Plot particle diameter (μm) versus particle relaxation time *)
```

```
data = {partDiam 10^4, tauPlusV}^T;
(* multiply particle diameter by 10^4 to convert from cm to μm *)
ListLogLogPlot[data,
 BaseStyle → {"Times New Roman", 12},
 Joined → True,
 Frame → True,
 FrameLabel → {"Particle diameter (μm)", "Particle relaxation time, τ⁺"},
 ImageSize → Large
]
```




```
(* Compute the non-dimensional deposition velocity as a function of particle diameter *)

vdPlusV = Table[vdPlus[partDiam[[i]], ustar], {i, 1, Length[partDiam]}];

(* Plot the non-dimensional deposition velocity vs the non-
 dimensional particle relaxation time *)
```

```
data = {tauPlusV, vdPlusV}^T;
ListLogLogPlot[data,
 BaseStyle → {"Times New Roman", 12},
 Joined → True,
 Frame → True,
 FrameLabel → {"Particle relaxation time, τ⁺", "Deposition velocity, v_d⁺"},
 ImageSize → Large
]
```